

## Between-Person and Within-Person Effects of Negative Mood Predicting Next-Morning Glucose

These data were simulated loosely based on real data reported in the citation below. The daily diary study followed persons with Type II diabetes for 21 consecutive days to examine within-person relationships between mood, stress, and morning glucose (an index of how well-controlled the diabetes is). Here we will examine between-person and within-person relationships between daily negative mood and glucose the next morning (which was log-transformed given skewness) and how these relationships are moderated by sex.

Skaff, M., Mullan J., Fisher, L., Almeida, D., **Hoffman, L.**, Masharani, U., & Mohr, D. (2009). Effects of mood on daily fasting glucose in Type 2 Diabetes. *Health Psychology, 28*(3), 265-272.

### SAS Data Setup:

```
* Read into work library and center predictors;
DATA work.Example3b; SET filesave.Example3b;
  * Level-2 effect of Negative Mood (mean=0, SD=1);
  pmnm0 = pmnegmood - 0; LABEL pmnm0 = "PM Negative Mood (0=0)";
  * Level-1 effect to use with PERSON-MEAN-CENTERING;
  wpmnm = negmood - pmnegmood; LABEL wpmnm = "WP Negative Mood (0=PM)";
  * Level-1 effect to use with GRAND-MEAN-CENTERING;
  tvnm0 = negmood - 0; LABEL tvnm0 = "TV Negative Mood (0=0)";
  * Gender already exists;
  LABEL sexMW = "Participant Sex(0=M, 1=W)";
RUN;
```

### PART 1: VARIANCE DECOMPOSITION

#### 1a) Empty Means, Random Intercept Model for Log-Transformed Next Morning Glucose (DV)

```
TITLE "Empty Means, Random Intercept Model for Glucose (TV Outcome)";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID Day;
  MODEL lGlucAM = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR SUBJECT=ID TYPE=UN;
  REPEATED Day / SUBJECT=ID TYPE=VC;
  ODS OUTPUT CovParms=CovEmpty InfoCrit=FitEmpty; * Save covparms, fit;
RUN;
```

$$\begin{aligned} \text{Level 1: } \text{Glucose}_{ti} &= \beta_{0i} + e_{ti} \\ \text{Level 2: } \beta_{0i} &= \gamma_{00} + U_{0i} \end{aligned}$$

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.06654	0.006690	9.95	<.0001
day	ID	0.03029	0.000683	44.35	<.0001

$$\text{ICC for Glucose (in VCORR): } .06654 / (.06654 + .03029) = .69$$

#### 1b) Empty Means, Random Intercept Model for Negative Mood (TV Predictor)

```
TITLE "Empty Means, Random Intercept Model for Negative Mood (TV Predictor)";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID Day;
  MODEL negmood = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR SUBJECT=ID TYPE=UN;
  REPEATED Day / SUBJECT=ID TYPE=VC; RUN;
```

$$\begin{aligned} \text{Level 1: } \text{Mood}_{ti} &= \beta_{0i} + e_{ti} \\ \text{Level 2: } \beta_{0i} &= \gamma_{00} + U_{0i} \end{aligned}$$

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.3355	0.03557	9.43	<.0001
day	ID	0.5258	0.01186	44.35	<.0001

$$\text{ICC for Negative Mood (in VCORR): } .3355 / (.3355 + .5258) = .39$$

**PART 2: PERSON-MEAN-CENTERING OF NEGATIVE MOOD TO PREDICT GLUCOSE****2a) Predicting Glucose from Fixed Effects of Negative Mood using Person-Mean-Centering:**

$$\text{Level 1: Glucose}_{i} = \beta_{0i} + \beta_{1i} (\text{Mood}_{i} - \overline{\text{Mood}}_{i}) + e_{i}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}}_{i} - 0) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10}$$

```

TITLE "Adding both fixed effects of negative mood under Person-MC";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
CLASS ID day;
MODEL lGlucAM = WPnm PMnm0 / SOLUTION DDFM=Satterthwaite OUTPM=PredMood;
RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
REPEATED day / SUBJECT=ID TYPE=VC;
ODS OUTPUT CovParms=CovMood InfoCrit=FitMood; * Save covparms, fit;
ESTIMATE "Within-Person Effect" WPnm 1;
ESTIMATE "Between-Person Effect" PMnm0 1;
ESTIMATE "Contextual Effect" WPnm -1 PMnm0 1;
RUN;

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	0.06435	0.006474	9.94	<.0001
Day	ID	0.03022	0.000682	44.35	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1956.5	5	-1946.5	-1946.5	-1939.8	-1929.9	-1924.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9302	0.01845	207	267.20	<.0001
WPnm	0.01097	0.003821	3941	2.87	0.0041
PMnm0	0.08040	0.03046	207	2.64	0.0089

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr >  t
Within-Person Effect	0.01097	0.003821	3941	2.87	0.0041
Between-Person Effect	0.08040	0.03046	207	2.64	0.0089
Contextual Effect	0.06942	0.03070	213	2.26	0.0247

What does the level-1 effect (WPnm) represent in this model?

What does the level-2 effect (PMnm0) represent in this model?

What does the "contextual effect" represent?

Which variance did the level-1 effect of WPnm account for?

Which variance did the level-2 effect of PMnm0 account for?

**\* Calculate PseudoR2 relative to empty model;**

```
%PseudoR2 (NCov=2, CovFewer=CovEmpty, CovMore=CovMood);
```

PseudoR2 (% Reduction) for CovEmpty vs. CovMood

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	ID	0.06654	0.006690	9.95	<.0001	.
CovEmpty	day	ID	0.03029	0.000683	44.35	<.0001	.
CovMood	UN(1,1)	ID	0.06435	0.006474	9.94	<.0001	0.032967
CovMood	day	ID	0.03022	0.000682	44.35	<.0001	0.002105

What is the total reduction in glucose variance so far?

```
PROC CORR DATA=PredMood; VAR lGlucAM pred; RUN;
```

Pearson Correlation Coefficients, N = 4140

Prob > |r| under H0: Rho=0

	lglucAM	Pred
lglucAM	1.00000	0.15269 <.0001

**Total variance accounted for in glucose by the effects of negative mood:  $r = .15269$ ,  $R^2 = .023$ .**

Is this total reduction in variance significant?

**\* Calculate difference in model fit relative to empty model;**

```
%FitTest(FitFewer=FitEmpty, FitMore=FitMood);
```

Likelihood Ratio Test for FitEmpty vs. FitMood

Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitEmpty	-1941.5	3	-1935.5	-1925.5	.	.	.
FitMood	-1956.5	5	-1946.5	-1929.9	15.0818	2	.000530910

## 2b) Testing a random effect of WP negative mood under Person-MC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - \overline{\text{Mood}}_i) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}}_i - 0) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + U_{1i}$$

```
TITLE "Add random effect of WP negative mood under Person-MC";
```

```
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
```

```
CLASS ID Day;
```

```
MODEL lGlucAM = WPnm PMnm0 / SOLUTION DDFM=Satterthwaite;
```

```
RANDOM INTERCEPT WPnm / SUBJECT=ID TYPE=UN;
```

```
REPEATED Day / SUBJECT=ID TYPE=VC;
```

```
ODS OUTPUT InfoCrit= FitRandMood; * Save fit;
```

```
ESTIMATE "Within-Person Effect" WPnm 1;
```

```
ESTIMATE "Between-Person Effect" PMnm0 1;
```

```
ESTIMATE "Contextual Effect" WPnm -1 PMnm0 1;
```

```
RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	0.06440	0.006479	9.94	<.0001
UN(2,1)	ID	-0.00020	0.001067	-0.19	0.8478
UN(2,2)	ID	0.000505	0.000335	1.51	0.0656
Day	ID	0.02995	0.000692	43.28	<.0001

		Information Criteria				
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1959.4	7	-1945.4	-1945.4	-1936.0	-1922.1	-1915.1

## Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9302	0.01846	207	267.10	<.0001
WPnm	0.01104	0.004137	202	2.67	0.0083
PMnm0	0.08022	0.03047	207	2.63	0.0091

Note the change in DF and SE for the now-random WPnm effect.

## Estimates

Label	Estimate	Standard Error	DF	t Value	Pr >  t
Within-Person Effect	0.01104	0.004137	202	2.67	0.0083
Between-Person Effect	0.08022	0.03047	207	2.63	0.0091
Contextual Effect	0.06918	0.03075	215	2.25	0.0255

Is this a better model than the fixed effects person-MC model (2a)? What does this result mean?

\* Calculate difference in model fit relative to fixed WPnm model;

```
%FitTest(FitFewer=FitMood, FitMore=FitRandMood);
```

Likelihood Ratio Test for FitMood vs. FitRandMood

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitMood	-1956.5	5	-1946.5	-1929.9	.	.	.
FitRandMood	-1959.4	7	-1945.4	-1922.1	2.90730	2	0.23372

## 2c) Adding moderation effects by sex (0=M, 1=W) for each mood effect under P-MC:

$$\text{Level 1: } \text{Glucose}_{it} = \beta_{0i} + \beta_{1i} (\text{Mood}_{it} - \overline{\text{Mood}_i}) + e_{it}$$

$$\text{Level 2: } \text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + \gamma_{02} (\text{Woman}_i) + \gamma_{03} (\overline{\text{Mood}_i} - 0)(\text{Woman}_i) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + \gamma_{12} (\text{Woman}_i)$$

```
TITLE "Add 3 effects of sex (0=M, 1=F) under Person-MC negative mood";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
CLASS ID Day;
MODEL lGlucAM = WPnm PMnm0 sexMW WPnm*sexMW PMnm0*sexMW
          / SOLUTION DDFM=Satterthwaite OUTPM=PredSex;
RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
REPEATED Day / SUBJECT=ID TYPE=VC;
ODS OUTPUT CovParms=CovSex InfoCrit=FitSex; * Save covparms, fit;
```

## Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	0.06074	0.006118	9.93	<.0001
Day	ID	0.03007	0.000678	44.35	<.0001

## Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1988.1	8	-1972.1	-1972.0	-1961.3	-1945.4	-1937.4

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9539	0.02734	207	181.21	<.0001
WPnm	0.03119	0.005937	3942	5.25	<.0001
PMnm0	0.1996	0.04849	207	4.12	<.0001
sexMW	-0.03619	0.03626	207	-1.00	0.3194
WPnm*sexMW	-0.03443	0.007743	3942	-4.45	<.0001
PMnm0*sexMW	-0.1849	0.06135	207	-3.01	0.0029

What does the intercept now represent in this model?

What does the level-1 effect (WPnm) represent in this model?

What does the level-2 effect (PMnm0) represent in this model?

What does the main effect of sex represent in this model?

What does the WPnm\*Sex interaction represent in this model?

What does the PMnm0\*Sex interaction represent in this model?

Which of these estimated effects were already given to us in the model?

Which of these estimated effects were NOT already given to us in the model?

```


ESTIMATE "Intercept: Men (Mood=0)"          intercept 1 sexMW 0;
ESTIMATE "Intercept: Women (Mood=0)"        intercept 1 sexMW 1;
ESTIMATE "Intercept: Women Diff (Mood=0)"    sexMW 1;

ESTIMATE "Within-Person Effect: Men"         WPnm 1 WPnm*sexMW 0;
ESTIMATE "Within-Person Effect: Women"      WPnm 1 WPnm*sexMW 1;
ESTIMATE "Within-Person Effect: Women Diff"  WPnm*sexMW 1;

ESTIMATE "Between-Person Effect: Men"        PMnm0 1 PMnm0*sexMW 0;
ESTIMATE "Between-Person Effect: Women"     PMnm0 1 PMnm0*sexMW 1;
ESTIMATE "Between-Person Effect: Women Diff" PMnm0*sexMW 1;

ESTIMATE "Contextual Effect: Men"           WPnm -1 WPnm*sexMW 0 PMnm0 1 PMnm0*sexMW 0;
ESTIMATE "Contextual Effect: Women"        WPnm -1 WPnm*sexMW -1 PMnm0 1 PMnm0*sexMW 1;
ESTIMATE "Contextual Effect: Women Diff"    WPnm*sexMW -1 PMnm0*sexMW 1;
RUN;
```

Label	Estimates		DF	t Value	Pr >  t
	Estimate	Standard Error			
Intercept: Men (Mood=0)	4.9539	0.02734	207	181.21	<.0001
Intercept: Women (Mood=0)	4.9177	0.02382	207	206.42	<.0001
Intercept: Women Diff (Mood=0)	-0.03619	0.03626	207	-1.00	0.3194
Within-Person Effect: Men	0.03119	0.005937	3942	5.25	<.0001
Within-Person Effect: Women	-0.00325	0.004970	3942	-0.65	0.5138
Within-Person Effect: Women Diff	-0.03443	0.007743	3942	-4.45	<.0001
Between-Person Effect: Men	0.1996	0.04849	207	4.12	<.0001
Between-Person Effect: Women	0.01469	0.03759	207	0.39	0.6962
Between-Person Effect: Women Diff	-0.1849	0.06135	207	-3.01	0.0029
Contextual Effect: Men	0.1684	0.04886	214	3.45	0.0007
Contextual Effect: Women	0.01794	0.03790	214	0.47	0.6364
Contextual Effect: Women Diff	-0.1505	0.06184	214	-2.43	0.0158



\* Calculate PseudoR2 relative to fixed-mood-only model;  
 %PseudoR2 (NCov=2, CovFewer=CovMood, CovMore=CovSex);

PseudoR2 (% Reduction) for CovMood vs. CovSex

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovMood	UN(1,1)	ID	0.06435	0.006474	9.94	<.0001	.
CovMood	day	ID	0.03022	0.000682	44.35	<.0001	.
CovSex	UN(1,1)	ID	0.06074	0.006118	9.93	<.0001	0.056080
CovSex	day	ID	0.03007	0.000678	44.35	<.0001	0.005027

Which new effects accounted for residual variance?

Which new effects accounted for random intercept variance?

What is the difference in the total reduction in glucose variance due to sex?

**\* Calculate Total R2 change relative to fixed-mood-only model;**

```
%TotalR2(DV=lGlucAM, PredFewer=PredMood, PredMore=PredSex);
```

Total R2 (% Reduction) for PredMood vs. PredSex

Name	Pred Corr	TotalR2	Total R2Diff
PredMood	0.15269	0.023315	.
PredSex	0.24931	0.062155	0.038840

Is this total new reduction in variance significant?

**\* Calculate difference in model fit relative to fixed-mood-only model;**

```
%FitTest(FitFewer=FitMood, FitMore=FitSex);
```

Likelihood Ratio Test for FitMood vs. FitSex

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitMood	-1956.5	5	-1946.5	-1929.9	.	.	.
FitSex	-1988.1	8	-1972.1	-1945.4	31.5122	3	.00000663

Had we used the CLASS statement for sex, here's what that code would have been instead.

Note that two values need to give for sex now because it is represented as two distinct groups (not a slope).

```
TITLE "SAS Person-MC Mood and Sex (0=M, 1=W) USING CLASS STATEMENT";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID day sexMW;
  MODEL lGlucAM = WPnm PMnm0 sexMW WPnm*sexMW PMnm0*sexMW
                / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
  ODS OUTPUT CovParms=CovSex InfoCrit=FitSex; * Save covparms, fit;
  ESTIMATE "Intercept: Men"          intercept 1 sexMW 1 0;
  ESTIMATE "Intercept: Women"       intercept 1 sexMW 0 1;
  ESTIMATE "Intercept: Women Diff"  sexMW -1 1;

  ESTIMATE "Within-Person Effect: Men"      WPnm 1 WPnm*sexMW 1 0;
  ESTIMATE "Within-Person Effect: Women"    WPnm 1 WPnm*sexMW 0 1;
  ESTIMATE "Within-Person Effect: Women Diff"  WPnm*sexMW -1 1;

  ESTIMATE "Between-Person Effect: Men"     PMnm0 1 PMnm0*sexMW 1 0;
  ESTIMATE "Between-Person Effect: Women"   PMnm0 1 PMnm0*sexMW 0 1;
  ESTIMATE "Between-Person Effect: Women Diff" PMnm0*sexMW -1 1;

  ESTIMATE "Context Effect: Men"    WPnm -1 WPnm*sexMW -1 0 PMnm0 1 PMnm0*sexMW 1 0;
  ESTIMATE "Context Effect: Women"  WPnm -1 WPnm*sexMW 0 -1 PMnm0 1 PMnm0*sexMW 0 1;
  ESTIMATE "Context Effect: Women Diff"  WPnm*sexMW 1 -1          PMnm0*sexMW -1 1;
RUN;
```

Person-mean-centering is one approach to including the effects of time-varying predictors.

Now let's examine the same series of models using the alternative approach—grand-mean-centering.

**\* Level-1 effect to use with GRAND-MEAN-CENTERING;**

```
tvnm0 = negmood - 0; LABEL tvnm0 = "TV Negative Mood (0=0)";
```

**PART 3: GRAND-MEAN-CENTERING OF NEGATIVE MOOD TO PREDICT GLUCOSE****3a) Predicting Glucose from Time-Varying Negative Mood only (Grand-MC):**

Level 1: $\text{Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_{ti}$ Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$ Time-Varying Mood: $\beta_{1i} = \gamma_{10}$
--

```

TITLE "Add fixed effect of level-1 negative mood only under Grand-MC";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
CLASS ID Day;
MODEL lGlucAM = TVnm0 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
REPEATED Day / SUBJECT=ID TYPE=VC;
ODS OUTPUT CovParms=CovSmush InfoCrit=FitSmush; * Save covparms, fit;
RUN;

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr >  Z
UN(1,1)	ID	0.06595	0.006634	9.94	<.0001
Day	ID	0.03022	0.000682	44.34	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1951.5	4	-1943.5	-1943.5	-1938.1	-1930.2	-1926.2

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9408	0.01806	207	273.52	<.0001
TVnm0	0.01202	0.003792	4041	3.17	0.0015

What does the level-1 effect of TVnm0 represent in this model?

How much variance did the level-1 effect of TVnm0 account for?

```

* Calculate PseudoR2 relative to empty model;
%PseudoR2 (NCov=2, CovFewer=CovEmpty, CovMore=CovSmush);

```

PseudoR2 (% Reduction) for CovEmpty vs. CovSmush							
Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	ID	0.06654	0.006690	9.95	<.0001	.
CovEmpty	day	ID	0.03029	0.000683	44.35	<.0001	.
CovSmush	UN(1,1)	ID	0.06595	0.006634	9.94	<.0001	.008842272
CovSmush	day	ID	0.03022	0.000682	44.34	<.0001	.002088088



**3b) Adding person mean negative mood at level 2 under Grand-MC (to un-smush the level-1 effect):**

Level 1: $\text{Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_{ti}$ Level 2: Intercept: $\beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i}$ Time-Varying Mood: $\beta_{1i} = \gamma_{10}$
---

```

TITLE "Add fixed effect of level-2 negative mood under Grand-MC";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID Day;
  MODEL lGlucAM = TVnm0 PMnm0 / SOLUTION DDFM=Satterthwaite OUTPM=PredTVMood;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED Day / SUBJECT=ID TYPE=VC;
  ODS OUTPUT CovParms=CovTVMood InfoCrit=FitTVMood; * Save covparms, fit;
  ESTIMATE "Within-Person Effect"      TVnm0 1;
  ESTIMATE "Between-Person Effect"     TVnm0 1 PMnm0 1;
  ESTIMATE "Contextual Effect"         PMnm0 1;          RUN;

```

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z	
UN(1,1)	ID	0.06435	0.006474	9.94	<.0001	
Day	ID	0.03022	0.000682	44.35	<.0001	

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1956.5	5	-1946.5	-1946.5	-1939.8	-1929.9	-1924.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9302	0.01845	207	267.20	<.0001
TVnm0	0.01097	0.003821	3941	2.87	0.0041
PMnm0	0.06942	0.03070	213	2.26	0.0247

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr >  t
Within-Person Effect	0.01097	0.003821	3941	2.87	0.0041
Between-Person Effect	0.08040	0.03046	207	2.64	0.0089
Contextual Effect	0.06942	0.03070	213	2.26	0.0247

What does the level-1 effect (TVnm0) NOW represent in this model?

What does the level-2 effect (PMnm0) represent in this model?

How much variance did the level-2 effect of PMnm0 account for?

```

* Calculate PseudoR2 relative to smushed model;
%PseudoR2 (NCov=2, CovFewer=CovSmush, CovMore=CovTVMood);

```

PseudoR2 (% Reduction) for CovSmush vs. CovTVMood							
Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovSmush	UN(1,1)	ID	0.06595	0.006634	9.94	<.0001	.
CovSmush	day	ID	0.03022	0.000682	44.34	<.0001	.
CovTVMood	UN(1,1)	ID	0.06435	0.006474	9.94	<.0001	0.024340
CovTVMood	day	ID	0.03022	0.000682	44.35	<.0001	0.000017

## 3c) Testing a random effect of TV negative mood under Grand-MC:

$$\begin{aligned} \text{Level 1: } \text{Glucose}_{ti} &= \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_{ti} \\ \text{Level 2: } \text{Intercept: } \beta_{0i} &= \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i} \\ \text{Time-Varying Mood: } \beta_{1i} &= \gamma_{10} + U_{1i} \end{aligned}$$

```
TITLE "Add random effect of level-1 negative mood only under Grand-MC";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID Day;
  MODEL lGlucAM = TVnm0 PMnm0 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT TVnm0 / SUBJECT=ID TYPE=UN;
  REPEATED Day / SUBJECT=ID TYPE=VC;
  ODS OUTPUT InfoCrit=FitTVRandMood; * Save fit;
  ESTIMATE "Within-Person Effect" TVnm0 1;
  ESTIMATE "Between-Person Effect" TVnm0 1 PMnm0 1;
  ESTIMATE "Contextual Effect" PMnm0 1;
RUN;
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	0.06400	0.006464	9.90	<.0001
UN(2,1)	ID	-0.00033	0.001050	-0.31	0.7549
UN(2,2)	ID	0.000579	0.000339	1.71	0.0441
Day	ID	0.02992	0.000690	43.34	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1960.4	7	-1946.4	-1946.4	-1937.0	-1923.1	-1916.1

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9302	0.01843	206	267.45	<.0001
TVnm0	0.01102	0.004181	205	2.64	0.0090
PMnm0	0.07015	0.03066	214	2.29	0.0231

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr >  t
Within-Person Effect	0.01102	0.004181	205	2.64	0.0090
Between-Person Effect	0.08117	0.03047	209	2.66	0.0083
Contextual Effect	0.07015	0.03066	214	2.29	0.0231

Is this a better model than the fixed effects grand-MC model (3b)? What does this result mean?

```
* Calculate difference in model fit relative to fixed-mood-only model;
%FitTest(FitFewer=FitTVMood, FitMore=FitTVRandMood);
```

Likelihood Ratio Test for FitTVMood vs. FitTVRandMood

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitTVMood	-1956.5	5	-1946.5	-1929.9	.	.	.
FitTVRandMood	-1960.4	7	-1946.4	-1923.1	3.85979	2	0.14516

Note that the PMC and GMC models no longer yield equivalent results if the level-1 effect is random.

## 3d) Adding main effect of sex and interactions with negative mood under Grand-MC:

$$\begin{aligned} \text{Level 1: } \text{Glucose}_{ti} &= \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_{ti} \\ \text{Level 2: } \text{Intercept: } \beta_{0i} &= \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + \gamma_{02} (\text{Woman}_i) + \gamma_{03} (\overline{\text{Mood}_i} - 0)(\text{Woman}_i) + U_{0i} \\ \text{Time-Varying Mood: } \beta_{1i} &= \gamma_{10} + \gamma_{12} (\text{Woman}_i) \end{aligned}$$

```
TITLE "Add sex, sex*TVnm0, and sex*PMnm0 under Grand-MC";
PROC MIXED DATA=work.Example3b COVTEST NOCLPRINT IC NAMELEN=100 METHOD=ML;
  CLASS ID Day;
  MODEL lGlucAM = TVnm0 PMnm0 sexMW TVnm0*sexMW PMnm0*sexMW
    / SOLUTION DDFM=Satterthwaite OUTPM=PredTVSex;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED Day / SUBJECT=ID TYPE=VC;
  ODS OUTPUT CovParms=CovTVSex InfoCrit=FitTVSex; * Save covparms, fit;
```

## Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr >  Z
UN(1,1)	ID	0.06074	0.006118	9.93	<.0001
Day	ID	0.03007	0.000678	44.35	<.0001

## Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1988.1	8	-1972.1	-1972.0	-1961.3	-1945.4	-1937.4

## Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr >  t
Intercept	4.9539	0.02734	207	181.21	<.0001
TVnm0	0.03119	0.005937	3942	5.25	<.0001
PMnm0	0.1684	0.04886	214	3.45	0.0007
sexMW	-0.03619	0.03626	207	-1.00	0.3194
TVnm0*sexMW	-0.03443	0.007743	3942	-4.45	<.0001
PMnm0*sexMW	-0.1505	0.06184	214	-2.43	0.0158

```
ESTIMATE "Intercept: Men"          intercept 1 sexMW 0;
ESTIMATE "Intercept: Women"       intercept 1 sexMW 1;
ESTIMATE "Intercept: Women Diff"  sexMW 1;

ESTIMATE "Within-Person Effect: Men" TVnm0 1 TVnm0*sexMW 0;
ESTIMATE "Within-Person Effect: Women" TVnm0 1 TVnm0*sexMW 1;
ESTIMATE "Within-Person Effect: Women Diff" TVnm0*sexMW 1;

ESTIMATE "Between-Person Effect: Men" TVnm0 1 TVnm0*sexMW 0 PMnm0 1 PMnm0*sexMW 0;
ESTIMATE "Between-Person Effect: Women" TVnm0 1 TVnm0*sexMW 1 PMnm0 1 PMnm0*sexMW 1;
ESTIMATE "Between-Person Effect: Women Diff" TVnm0*sexMW 1 PMnm0*sexMW 1;

ESTIMATE "Contextual Effect: Men"    PMnm0 1 PMnm0*sexMW 0;
ESTIMATE "Contextual Effect: Women"  PMnm0 1 PMnm0*sexMW 1;
ESTIMATE "Contextual Effect: Women Diff" PMnm0*sexMW 1;
RUN;
```

Label	Estimates		DF	t Value	Pr >  t
	Estimate	Standard Error			
Intercept: Men	4.9539	0.02734	207	181.21	<.0001
Intercept: Women	4.9177	0.02382	207	206.42	<.0001
Intercept: Women Diff	-0.03619	0.03626	207	-1.00	0.3194
Within-Person Effect: Men	0.03119	0.005937	3942	5.25	<.0001
Within-Person Effect: Women	-0.00325	0.004970	3942	-0.65	0.5138
Within-Person Effect: Women Diff	-0.03443	0.007743	3942	-4.45	<.0001
Between-Person Effect: Men	0.1996	0.04849	207	4.12	<.0001
Between-Person Effect: Women	0.01469	0.03759	207	0.39	0.6962
Between-Person Effect: Women Diff	-0.1849	0.06135	207	-3.01	0.0029
Contextual Effect: Men	0.1684	0.04886	214	3.45	0.0007
Contextual Effect: Women	0.01794	0.03790	214	0.47	0.6364
Contextual Effect: Women Diff	-0.1505	0.06184	214	-2.43	0.0158

\* Calculate PseudoR2 relative to fixed-mood-only model;  
 %PseudoR2 (NCov=2, CovFewer=CovTVMood, CovMore=CovTVSex);

PsuedoR2 (% Reduction) for CovTVMood vs. CovTVSex

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovTVMood	UN(1,1)	ID	0.06435	0.006474	9.94	<.0001	.
CovTVMood	day	ID	0.03022	0.000682	44.35	<.0001	.
CovTVSex	UN(1,1)	ID	0.06074	0.006118	9.93	<.0001	0.056080
CovTVSex	day	ID	0.03007	0.000678	44.35	<.0001	0.005027

Which new effects accounted for residual variance?

Which new effects accounted for random intercept variance?

What is the difference in the total reduction in glucose variance due to sex?

\* Calculate Total R2 change relative to fixed-mood-only model;  
 %TotalR2 (DV=lGlucAM, PredFewer=PredTVMood, PredMore=PredTVSex);

Total R2 (% Reduction) for PredTVMood vs. PredTVSex

Name	Pred		Total	
	Corr	TotalR2	R2Diff	
PredTVMood	0.15269	0.023315	.	
PredTVSex	0.24931	0.062155	0.038840	

Is this total new reduction in variance significant?

\* Calculate difference in model fit relative to fixed-mood-only model;  
 %FitTest (FitFewer=FitTVMood, FitMore=FitTVSex);

Likelihood Ratio Test for FitTVMood vs. FitTVSex

Name	Neg2Log		AIC	BIC	DevDiff	DFdiff	Pvalue
	Like	Parms					
FitTVMood	-1956.5	5	-1946.5	-1929.9	.	.	.
FitTVSex	-1988.1	8	-1972.1	-1945.4	31.5122	3	.000000663

Sample Results Section (note the order of the models is different than what is in the handout):

The effects of negative mood and sex on next day's morning glucose level were examined in 207 persons with type-2 diabetes over a 20-day period. Glucose was natural log transformed (after adding 1 to each score) to improve normality. Intraclass correlations as calculated from an empty means, random intercept only model were .69 for glucose and .39 for negative mood, such that 69% and 39% of the variance in each variable was between persons, respectively. Preliminary analyses suggested that a random intercept only model for the variances of glucose over time had acceptable fit, and thus all conditional (predictor) models were examined using that structure as a baseline.

The time-varying (level-1) predictor for negative mood (left uncentered, given that 0 represented average level of the measure) was first entered into the model. A significant positive effect was obtained, such that higher daily levels of negative mood were related to higher daily levels of glucose. However, the inclusion of a single parameter for the effect of negative mood presumes that its between-person and within-person effects would be equivalent. This convergence hypothesis was tested explicitly by including person mean negative mood (also left uncentered, given that 0 represented average level of the original measure) as a level-2 predictor. The effect of person mean negative mood was significant, indicating that after controlling for absolute level of daily negative mood, persons with higher mean negative mood had higher mean glucose. Given that the significance of the level-2 effect also indicates that the between-person and within-person effects of negative mood were not equivalent, the model was re-specified to facilitate interpretation of these separate effects using group-mean-centering (i.e., person-mean-centering in longitudinal data). Specifically, a new level-1 predictor variable was created by subtracting each person's mean from daily negative mood, while the level-2 effect continued to be represented by the person mean. In this specification using person-mean-centering, the level-2 mean of negative mood represents the between-person effect directly and the level-1 within-person deviation of negative mood represents the within-person effect directly. Both the between- and within-person effects of negative mood were significantly positive. A random level-1 effect of negative mood was tested within both models, and was not found to be significant in either,  $-2\Delta LL (\sim 2) < 5.14, p > .05$ , indicating no significant individual differences in the within-person effect of negative mood.

Three effects of sex were then entered into the person-mean-centered model, including a main effect of sex and interactions with the between- and within-person effects of negative mood. The main effect of sex was non-significant, indicating no sex differences in mean glucose among persons with average levels of mean negative mood on average days (i.e., when average persons were at their mean). Given that both interactions were significant, however, results for both men and women will be presented as derived from ESTIMATE statements for the effects estimated specifically for each group within the overall model. Parameters for this final model are given in Table 1.

As shown, the intercept of 4.95 represents the expected morning LN glucose for a man with an average level of mean negative mood on an average day (i.e., both mean and person-mean-centered negative mood at 0). Men showed significant between- and within-person effects of negative mood, such that for every unit higher in mean negative mood, mean glucose was expected to be 0.20 higher (i.e., the between-person effect), and for every unit higher in negative mood on a given day relative to his own mean, glucose that next morning was expected to be 0.03 higher as well (i.e., the within-person effect). Thus, in men, being higher overall in negative mood and higher than usual in negative mood were each related to higher levels of glucose, and these effects were significantly different in magnitude (contextual effect = 0.17, SE = 0.05,  $p < .001$ ). Said differently the contextual effect also indicates a significant contribution of person mean negative mood after controlling for daily negative mood.

As shown in Figure 1, however, these patterns were not found in women, as indicated by the significant interactions with sex. Specifically, the between-person and within-person effects of negative mood in women were 0.015 (SE = 0.038) and  $-0.003$  (SE = 0.005), respectively. Neither effect was significant nor did they differ significantly in magnitude (contextual effect = 0.018, SE = .038). Both effects of negative mood were significantly smaller than in men (interaction terms of sex with between-person and within-person negative mood of  $-0.185$  and  $-0.034$ , respectively). Finally, the contextual effect of negative mood, or the difference between the between-person and within-person effects of negative mood, was significantly larger for men (0.151, SE = 0.062,  $p = .016$ ).

(Table 1 would have all parameter estimates from final model, see chapter 8 for examples)

(Figure 1 would show the within-person effect of negative mood for men and women with low or high mean negative mood – see plot for an example)